Pensieve header: Γ-Calculus.

```
<< KnotTheory`
```

```
Loading KnotTheory` version of September 6, 2014, 13:37:37.2841.
Read more at http://katlas.org/wiki/KnotTheory.
```

Linear Control Theory

From the Wikipedia entry “Control Theory” (edited):

Closed-loop transfer function [edit]

This is called a single-input-single-output (*SISO*) control system; *MIMO* (i.e., Multi-Input-Multi-Output) systems, with more than one input/output, are common. In such cases variables are represented through vectors instead of simple scalar values. For some distributed parameter systems the vectors may be infinite-dimensional (typically functions).



Solving for *Y*(*s*) in terms of *R*(*s*) gives

$$Y(s)=\left(\frac{P(s)C(s)}{1+F(s)P(s)C(s)}\right)R(s)=H(s)R(s).$$

```
M = (α β Θ
     γ δ ϵ
     ϕ ψ Ξ); {y1, y2, y3} == M.{x1, x2, x3}
```

$\{y_1, y_2, y_3\} == \{\alpha x_1 + \beta x_2 + \Theta x_3,\ \gamma x_1 + \delta x_2 + \epsilon x_3,\ \phi x_1 + \psi x_2 + \Xi x_3\}$

```
eqns = Thread[{y1, y2, y3} == M.{x1, x2, x3}] ⋃ {y1 == x2}
```

$\{y_1 == x_2,\ y_1 == \alpha x_1 + \beta x_2 + \Theta x_3,\ y_2 == \gamma x_1 + \delta x_2 + \epsilon x_3,\ y_3 == \phi x_1 + \psi x_2 + \Xi x_3\}$

```
Solve[eqns, {y2, y3}]
```

{}

```
{sol} = Solve[eqns, {x2, y1, y2, y3}]
```

$$\left\{\left\{x_2 \to -\frac{\alpha x_1 + \Theta x_3}{-1+\beta},\ y_1 \to -\frac{\alpha x_1 + \Theta x_3}{-1+\beta},\ y_2 \to -\frac{\gamma x_1 - \beta\gamma x_1 + \alpha\delta x_1 + \epsilon x_3 - \beta\epsilon x_3 + \delta\Theta x_3}{-1+\beta},\ y_3 \to -\frac{\phi x_1 - \beta\phi x_1 + \alpha\psi x_1 + \Xi x_3 - \beta\Xi x_3 + \Theta\psi x_3}{-1+\beta}\right\}\right\}$$

```
Table[Coefficient[yi /. sol, xj] // Apart, {i, {2, 3}}, {j, {1, 3}}] // MatrixForm
```

$$\begin{pmatrix} \gamma - \frac{\alpha\delta}{-1+\beta} & \epsilon - \frac{\delta\Theta}{-1+\beta} \\ \phi - \frac{\alpha\psi}{-1+\beta} & \Xi - \frac{\Theta\psi}{-1+\beta} \end{pmatrix}$$

Γ0-Calculus

```
Γ0 /: Collect[Γ0[λ_]] := Γ0[Collect[λ, h_, Collect[#, t_, Factor] &]];
Format[Γ0[λ_]] := Module[{S, M},
   S = Union@Cases[Γ0[λ], (h | t)a_ :> a, ∞];
   M = Outer[Factor[∂h#1t#2 λ] &, S, S];
   M = Prepend[M, t# & /@ S] // Transpose;
   M = Prepend[M, Prepend[h# & /@ S, "Γ0"]];
   M // MatrixForm];
```

$$\zeta = \Gamma0\left[\{t_1, t_2, t_3, t_S\}.\begin{pmatrix} \alpha_{11} & \alpha_{12} & \alpha_{13} & \theta_1 \\ \alpha_{21} & \alpha_{22} & \alpha_{23} & \theta_2 \\ \alpha_{31} & \alpha_{32} & \alpha_{33} & \theta_3 \\ \phi_1 & \phi_2 & \phi_3 & \Xi \end{pmatrix}.\{h_1, h_2, h_3, h_S\}\right]$$

$$\begin{pmatrix} \Gamma0 & h_1 & h_2 & h_3 & h_S \\ t_1 & \alpha_{11} & \alpha_{12} & \alpha_{13} & \theta_1 \\ t_2 & \alpha_{21} & \alpha_{22} & \alpha_{23} & \theta_2 \\ t_3 & \alpha_{31} & \alpha_{32} & \alpha_{33} & \theta_3 \\ t_S & \phi_1 & \phi_2 & \phi_3 & \Xi \end{pmatrix}$$

$$\Gamma0\ /:\ \Gamma0[\lambda 1\_]\ \Gamma0[\lambda 2\_] := \Gamma0[\lambda 1 + \lambda 2];$$

$$m_{a\_,b\_\to c\_}[\Gamma0[\lambda\_]] := \text{Module}\Big[\{\alpha, \beta, \gamma, \delta, \theta, \epsilon, \phi, \psi, \Xi\},$$

$$\begin{pmatrix} \alpha & \beta & \theta \\ \gamma & \delta & \epsilon \\ \phi & \psi & \Xi \end{pmatrix} = \begin{pmatrix} \partial_{t_a,h_a}\lambda & \partial_{t_a,h_b}\lambda & \partial_{t_a}\lambda \\ \partial_{t_b,h_a}\lambda & \partial_{t_b,h_b}\lambda & \partial_{t_b}\lambda \\ \partial_{h_a}\lambda & \partial_{h_b}\lambda & \lambda \end{pmatrix}\ /.\ (t \mid h)_{a|b} \to 0;$$

$$\Gamma0\left[\{t_c, 1\}.\begin{pmatrix} \gamma - \frac{\alpha\,\delta}{-1+\beta} & \epsilon - \frac{\delta\,\theta}{-1+\beta} \\ \phi - \frac{\alpha\,\psi}{-1+\beta} & \Xi - \frac{\theta\,\psi}{-1+\beta} \end{pmatrix}.\{h_c, 1\}\right]\ //\ \text{Collect}\Big];$$

$$(\zeta\ //\ m_{12\to 1}\ //\ m_{13\to 1})$$

$$\begin{pmatrix} \Gamma0 & h_1 & h_S \\ t_1 & \frac{\alpha_{31}-\alpha_{12}\alpha_{31}-\alpha_{13}\alpha_{22}\alpha_{31}-\alpha_{23}\alpha_{31}+\alpha_{12}\alpha_{23}\alpha_{31}+\alpha_{11}\alpha_{32}+\alpha_{13}\alpha_{21}\alpha_{32}-\alpha_{11}\alpha_{23}\alpha_{32}+\alpha_{21}\alpha_{33}-\alpha_{12}\alpha_{21}\alpha_{33}+\alpha_{11}\alpha_{22}\alpha_{33}}{1-\alpha_{12}-\alpha_{13}\alpha_{22}-\alpha_{23}+\alpha_{12}\alpha_{23}} & \frac{\alpha_{32}\theta_1-\alpha_{23}\alpha_{32}\theta_1+\alpha_{22}\alpha_{33}\theta_1+\alpha_{13}\alpha_{32}\theta_2+\alpha_{33}\theta_2-\alpha_{12}\alpha_{33}\theta_2+\theta_3-c}{1-\alpha_{12}-\alpha_{13}\alpha_{22}-\alpha_{23}+\alpha_{12}\alpha_{23}} \\ t_S & \frac{\phi_1-\alpha_{12}\phi_1-\alpha_{13}\alpha_{22}\phi_1-\alpha_{23}\phi_1+\alpha_{12}\alpha_{23}\phi_1+\alpha_{11}\phi_2+\alpha_{13}\alpha_{21}\phi_2-\alpha_{11}\alpha_{23}\phi_2+\alpha_{21}\phi_3-\alpha_{12}\alpha_{21}\phi_3+\alpha_{11}\alpha_{22}\phi_3}{1-\alpha_{12}-\alpha_{13}\alpha_{22}-\alpha_{23}+\alpha_{12}\alpha_{23}} & \frac{\Xi-\Xi\alpha_{12}-\Xi\alpha_{13}\alpha_{22}-\Xi\alpha_{23}+\Xi\alpha_{12}\alpha_{23}+\theta_1\phi_2-\alpha_{23}\theta_1\phi_2+\alpha_{13}\theta_2}{1-\alpha_{12}-\alpha_{13}\alpha_{22}-\alpha_{23}+\alpha_{12}\alpha_{23}} \end{pmatrix}$$

$$(\zeta\ //\ m_{12\to 1}\ //\ m_{13\to 1}) == (\zeta\ //\ m_{23\to 2}\ //\ m_{12\to 1})$$

True

The YB Elements for Γ0-Calculus

$$R_{n\_,i\_,j\_} := \text{ReplacePart}[\text{IdentityMatrix}[n], \{\{i, i\} \to \alpha, \{i, j\} \to \beta, \{j, i\} \to \gamma, \{j, j\} \to \delta\}];$$

$$\text{MatrixForm}\ /@\ \{R_{2,1,2}, R_{3,1,3}\}$$

$$\left\{\begin{pmatrix} \alpha & \beta \\ \gamma & \delta \end{pmatrix}, \begin{pmatrix} \alpha & 0 & \beta \\ 0 & 1 & 0 \\ \gamma & 0 & \delta \end{pmatrix}\right\}$$

$$R_{3,1,2}.R_{3,1,3}.R_{3,2,3}$$

$$\{\{\alpha^2, \alpha\beta + \alpha\beta\gamma, \beta^2 + \alpha\beta\delta\}, \{\alpha\gamma, \beta\gamma^2 + \alpha\delta, \beta\delta + \beta\gamma\delta\}, \{\gamma, \gamma\delta, \delta^2\}\}$$

$$\text{eqns} = \text{Thread}[\text{Flatten}[R_{3,1,2}.R_{3,1,3}.R_{3,2,3}] == \text{Flatten}[R_{3,2,3}.R_{3,1,3}.R_{3,1,2}]]$$

$$\{\text{True}, \alpha\beta + \alpha\beta\gamma == \alpha\beta, \beta^2 + \alpha\beta\delta == \beta, \alpha\gamma == \alpha\gamma + \alpha\beta\gamma, \beta\gamma^2 + \alpha\delta == \beta^2\gamma + \alpha\delta, \beta\delta + \beta\gamma\delta == \beta\delta, \gamma == \gamma^2 + \alpha\gamma\delta, \gamma\delta == \gamma\delta + \beta\gamma\delta, \text{True}\}$$

$$\text{Solve}[\text{eqns}, \{\alpha, \beta, \gamma, \delta\}]$$

Solve: Equations may not give solutions for all "solve" variables.

$$\{\{\beta \to 0, \gamma \to 0\}, \{\beta \to 0, \gamma \to 1 - \alpha\delta\}, \{\beta \to 1 - \alpha\delta, \gamma \to 0\}, \{\alpha \to 0, \beta \to 1, \gamma \to 1, \delta \to 0\}\}$$

$$\text{MatrixForm}\ /@\ (R_{2,1,2}\ /.\ \text{Solve}[\text{eqns}, \{\alpha, \beta, \gamma, \delta\}])$$

Solve: Equations may not give solutions for all "solve" variables.

$$\left\{\begin{pmatrix} \alpha & 0 \\ 0 & \delta \end{pmatrix}, \begin{pmatrix} \alpha & 0 \\ 1 - \alpha\delta & \delta \end{pmatrix}, \begin{pmatrix} \alpha & 1 - \alpha\delta \\ 0 & \delta \end{pmatrix}, \begin{pmatrix} 0 & 1 \\ 1 & 0 \end{pmatrix}\right\}$$

```
MatrixForm /@ (R2,1,2 /. Solve[eqns ⋃ {α == 0, δ == t}, {α, β, γ, δ}])
```

$$\left\{\begin{pmatrix} 0 & 0 \\ 1 & t \end{pmatrix}, \begin{pmatrix} 0 & 1 \\ 0 & t \end{pmatrix}, \begin{pmatrix} 0 & 0 \\ 0 & t \end{pmatrix}\right\}$$

```
MatrixForm /@ (R2,1,2 /. Solve[eqns ⋃ {α == 1, δ == t}, {α, β, γ, δ}])
```

$$\left\{\begin{pmatrix} 1 & 0 \\ 1-t & t \end{pmatrix}, \begin{pmatrix} 1 & 1-t \\ 0 & t \end{pmatrix}, \begin{pmatrix} 1 & 0 \\ 0 & t \end{pmatrix}\right\}$$

```
Γ0[Xp_{a_,b_} * more_.] := Γ0[{t_a, t_b}.(1  1-T; 0  T).{h_a, h_b}] Γ0[more];
Γ0[Xm_{a_,b_} * more_.] := (Γ0[Xp_{ab}] /. T → 1/T) Γ0[more];
Γ0[1] = Γ0[0];
```




```
{Xm51 Xm62 Xp34 // Γ0 // m14→1 // m25→2 // m36→3,  Xp61 Xm24 Xm35 // Γ0 // m14→1 // m25→2 // m36→3}
```

$$\left\{\begin{pmatrix} \Gamma 0 & h_1 & h_2 & h_3 \\ t_1 & 1 & 0 & 0 \\ t_2 & \frac{-1+T}{T} & \frac{1}{T} & 0 \\ t_3 & -\frac{-1+T}{T} & \frac{-1+T}{T} & 1 \end{pmatrix}, \begin{pmatrix} \Gamma 0 & h_1 & h_2 & h_3 \\ t_1 & 1 & 0 & 0 \\ t_2 & \frac{-1+T}{T} & \frac{1}{T} & 0 \\ t_3 & -\frac{-1+T}{T} & \frac{-1+T}{T} & 1 \end{pmatrix}\right\}$$

Two types of R2!

```
{Xp12 Xm34 // Γ0 // m13→1 // m24→2, Xp12 Xm34 // Γ0 // m13→1 // m42→2}
```

$$\left\{\begin{pmatrix} \Gamma 0 & h_1 & h_2 \\ t_1 & 1 & 0 \\ t_2 & 0 & 1 \end{pmatrix}, \begin{pmatrix} \Gamma 0 & h_1 & h_2 \\ t_1 & 1 & 0 \\ t_2 & 0 & 1 \end{pmatrix}\right\}$$




```
z = Γ0[Xm12,1 Xm27 Xm83 Xm4,11 Xp16,5 Xp6,13 Xp14,9 Xp10,15];
Do[z = z // m1k→1, {k, 2, 16}]; z
```

$$\begin{pmatrix} \Gamma 0 & h_1 \\ t_1 & 1 \end{pmatrix}$$

Γ-Calculus

```
Γ /: Collect[Γ[ω_, λ_]] := Γ[Simplify[ω],
   Collect[λ, h_, Collect[#, t_, Factor] &]];
Format[Γ[ω_, λ_]] := Module[{S, M},
   S = Union@Cases[Γ[ω, λ], (h | t)_a_ ⧴ a, ∞];
   M = Outer[Factor[∂_{h_#1 t_#2} λ] &, S, S];
   M = Prepend[M, t_# & /@ S] // Transpose;
   M = Prepend[M, Prepend[h_# & /@ S, ω]];
   M // MatrixForm];
```

```
ξ = Γ[ω, {t1, t2, t3, tS}.{{α11, α12, α13, θ1}, {α21, α22, α23, θ2}, {α31, α32, α33, θ3}, {ϕ1, ϕ2, ϕ3, Ξ}}.{h1, h2, h3, hS}]
```

$$\begin{pmatrix} \omega & h_1 & h_2 & h_3 & h_S \\ t_1 & \alpha_{11} & \alpha_{12} & \alpha_{13} & \theta_1 \\ t_2 & \alpha_{21} & \alpha_{22} & \alpha_{23} & \theta_2 \\ t_3 & \alpha_{31} & \alpha_{32} & \alpha_{33} & \theta_3 \\ t_S & \phi_1 & \phi_2 & \phi_3 & \Xi \end{pmatrix}$$

```
Γ /: Γ[ω1_, λ1_] Γ[ω2_, λ2_] := Γ[ω1 * ω2, λ1 + λ2];
m_{a_ b_ → c_}[Γ[ω_, λ_]] := Module[{α, β, γ, δ, θ, ϵ, ϕ, ψ, Ξ, μ},
   {{α, β, θ}, {γ, δ, ϵ}, {ϕ, ψ, Ξ}} =
     {{∂_{t_a,h_a} λ, ∂_{t_a,h_b} λ, ∂_{t_a} λ},
      {∂_{t_b,h_a} λ, ∂_{t_b,h_b} λ, ∂_{t_b} λ},
      {∂_{h_a} λ, ∂_{h_b} λ, λ}} /. (t | h)_{a|b} → 0;
   Γ[(μ = 1 - β) ω, {t_c, 1}.{{γ + α δ / μ, ϵ + δ θ / μ}, {ϕ + α ψ / μ, Ξ + ψ θ / μ}}.{h_c, 1}] // Collect];
Γ[Xp_{a_ b_} * more_.] := Γ[1, {t_a, t_b}.{{1, 1 - T}, {0, T}}.{h_a, h_b}] Γ[more];
Γ[Xm_{a_ b_} * more_.] := (Γ[Xp_{ab}] /. T → 1/T) Γ[more];
Γ[1] = Γ[1, 0];
```

```
(ξ // m12→1 // m13→1)
```

$$\begin{pmatrix} \omega\left(1-\alpha_{13}\alpha_{22}+\alpha_{12}\left(-1+\alpha_{23}\right)-\alpha_{23}\right) & h_1 & \cdots \\ t_1 & \frac{\alpha_{31}-\alpha_{12}\alpha_{31}-\alpha_{13}\alpha_{22}\alpha_{31}-\alpha_{23}\alpha_{31}+\alpha_{12}\alpha_{23}\alpha_{31}+\alpha_{11}\alpha_{32}+\alpha_{13}\alpha_{21}\alpha_{32}-\alpha_{11}\alpha_{23}\alpha_{32}+\alpha_{21}\alpha_{33}-\alpha_{12}\alpha_{21}\alpha_{33}+\alpha_{11}\alpha_{22}\alpha_{33}}{1-\alpha_{12}-\alpha_{13}\alpha_{22}-\alpha_{23}+\alpha_{12}\alpha_{23}} & \alpha_{32}\theta_1-\alpha_{23}\alpha_{32}\cdots \\ t_S & \frac{\phi_1-\alpha_{12}\phi_1-\alpha_{13}\alpha_{22}\phi_1-\alpha_{23}\phi_1+\alpha_{12}\alpha_{23}\phi_1+\alpha_{11}\phi_2+\alpha_{13}\alpha_{21}\phi_2-\alpha_{11}\alpha_{23}\phi_2+\alpha_{21}\phi_3-\alpha_{12}\alpha_{21}\phi_3+\alpha_{11}\alpha_{22}\phi_3}{1-\alpha_{12}-\alpha_{13}\alpha_{22}-\alpha_{23}+\alpha_{12}\alpha_{23}} & \Xi-\Xi\alpha_{12}-\cdots \end{pmatrix}$$

```
(ξ // m12→1 // m13→1) == (ξ // m23→2 // m12→1)
```

True




```
{Xm51 Xm62 Xp34 // Γ // m14→1 // m25→2 // m36→3, Xp61 Xm24 Xm35 // Γ // m14→1 // m25→2 // m36→3}
```

$$\left\{\begin{pmatrix} 1 & h_1 & h_2 & h_3 \\ t_1 & 1 & 0 & 0 \\ t_2 & \frac{-1+T}{T} & \frac{1}{T} & 0 \\ t_3 & -\frac{-1+T}{T} & \frac{-1+T}{T} & 1 \end{pmatrix}, \begin{pmatrix} 1 & h_1 & h_2 & h_3 \\ t_1 & 1 & 0 & 0 \\ t_2 & \frac{-1+T}{T} & \frac{1}{T} & 0 \\ t_3 & -\frac{-1+T}{T} & \frac{-1+T}{T} & 1 \end{pmatrix}\right\}$$

Two types of R2.

```
{Xp12 Xm34 // Γ // m13→1 // m24→2, Xp12 Xm34 // Γ // m13→1 // m42→2}
```

$$\left\{\begin{pmatrix} 1 & h_1 & h_2 \\ t_1 & 1 & 0 \\ t_2 & 0 & 1 \end{pmatrix}, \begin{pmatrix} 1 & h_1 & h_2 \\ t_1 & 1 & 0 \\ t_2 & 0 & 1 \end{pmatrix}\right\}$$




```
z = Γ[Xm12,1 Xm27 Xm83 Xm4,11 Xp16,5 Xp6,13 Xp14,9 Xp10,15];
Do[z = z // m1k→1, {k, 2, 16}]; z
```

$$\begin{pmatrix} 11 - \frac{1}{T^3} + \frac{4}{T^2} - \frac{8}{T} - 8T + 4T^2 - T^3 & h_1 \\ t_1 & 1 \end{pmatrix}$$

```
Alexander[Knot[8, 17]][t]
```

KnotTheory: Loading precomputed data in PD4Knots`.

$$11 - \frac{1}{t^3} + \frac{4}{t^2} - \frac{8}{t} - 8t + 4t^2 - t^3$$

Dror Bar-Natan: Classes: 2017: MAT 1350 AKT:
Handout for January 31, 2017:

Γ-Calculus

Derived from Cheat Sheet Meta-Calculi, in http://drorbn.net/AcademicPensieve/Projects/MetaCalculi/.

σ-calculus. $\sigma_1 * \sigma_2 = \sigma_1 \cup \sigma_2$, $m_c^{ab}(\sigma) = (\sigma \setminus \{a,b\}) \cup (c \to \sigma_a\sigma_b) \,/\, (T_a, T_b \to T_c)$, $\mathrm{tr}_c(\sigma) = \sigma \setminus c$, $R_{ab}^{\pm} \mapsto (a \to 1,\, b \to T_a^{\pm 1})$

Gassner calculus / Γ-calculus. Preserves $C_1 :=$ [col sum = 1] (⇔OC) and ✓ $C_2 := [\forall a, b,\ (T_a - 1) \mid (A_{ab} - \delta_{ab}\sigma_b)]$
• Except under tr_c, at $T_* = 1$, $\omega = 1$ and $A = I$.

$$\begin{array}{c|ccc} \omega & a & b & S \\ \hline a & \alpha & \beta & \theta \\ b & \gamma & \delta & \epsilon \\ S & \phi & \psi & \Xi \end{array} \xrightarrow[\substack{\mu := 1-\beta \\ T_a, T_b \to T_c}]{m_c^{ab}} \begin{array}{c|cc} \mu\omega & c & S \\ \hline c & \gamma + \alpha\delta/\mu & \epsilon + \delta\theta/\mu \\ S & \phi + \alpha\psi/\mu & \Xi + \psi\theta/\mu \end{array}$$

$$\begin{array}{c|cc} \omega & c & S \\ \hline c & \alpha & \theta \\ S & \psi & \Xi \end{array} \xrightarrow[\mu := 1-\alpha]{\mathrm{tr}_c} \begin{array}{c|c} \mu\omega & S \\ \hline S & \Xi + \psi\theta/\mu \end{array} \qquad R_{ab}^{\pm} \underset{\Gamma}{=} \begin{array}{c|cc} 1 & a & b \\ \hline a & 1 & 1 - T_a^{\pm 1} \\ b & 0 & T_a^{\pm 1} \end{array}$$

$$\begin{array}{c|cc} \omega & a & S \\ \hline a & \alpha & \theta \\ S & \phi & \Xi \end{array} \xrightarrow[\substack{\mu := T_a - 1 \\ \nu := \alpha - \sigma_a}]{\Delta_{bc}^a} \left(\begin{array}{c|ccc} \omega & b & c & S \\ \hline b & (\sigma_a - \alpha T_a - \nu T_c)/\mu & (T_b - 1)T_c\nu/\mu & (T_b - 1)T_c\theta/\mu \\ c & (T_c - 1)\nu/\mu & (\alpha - \sigma_a T_a - \nu T_c)/\mu & (T_c - 1)\theta/\mu \\ S & \phi & \phi & \Xi \end{array}\right)_{T_a \mapsto T_b T_c}$$

Satisfies: ✓ $R_{13}^{+} /\!\!/ \Delta_{12}^{1} = R_{23}^{+} \# R_{13}^{+}$.
✓ $R_{13}^{-} /\!\!/ \Delta_{12}^{1} = R_{13}^{-} \# R_{23}^{-}$.
✓ $\Delta_{a_1 a_2}^{a} /\!\!/ \Delta_{b_1 b_2}^{b} /\!\!/ m_{c_1}^{a_1 b_1} /\!\!/ m_{c_2}^{a_2 b_2} = m_c^{ab} /\!\!/ \Delta_{c_1 c_2}^{c}$.

$$\begin{array}{c|cc} \omega & a & S \\ \hline a & \alpha & \theta \\ S & \phi & \Xi \end{array} \xrightarrow{S^a} \left(\begin{array}{c|cc} \alpha\omega/\sigma_a & a & S \\ \hline a & 1/\alpha & \theta/\alpha \\ S & -\phi/\alpha & (\alpha\Xi - \phi\theta)/\alpha \end{array}\right)_{T_a \to T_a^{-1}}$$

Satisfies: ✓ $R_{12}^{\pm} /\!\!/ S^{1 \text{ or } 2} = R_{12}^{\mp}$. ✓ $m_c^{ab} /\!\!/ S^c = S^a /\!\!/ S^b /\!\!/ m_c^{ba}$.
✓ $S^a /\!\!/ S^a = I$. ✓ $\Delta_{bc}^{a} /\!\!/ S^b /\!\!/ S^c = S^a /\!\!/ \Delta_{cb}^{a}$.
✓ Assuming C_2, $\eta^a /\!\!/ \epsilon_a = \Delta_{bc}^{a} /\!\!/ S^c /\!\!/ m_a^{bc}$ (also 3 variants).

The map (tangle $T \mapsto$ matrix A) is anti-multiplicative. The MVA mod units: $L \mapsto (\omega, A) \mapsto \omega \det'(A - I)/(1 - T')$ ✓